' Calculate the price of a bond using yield-to-maturity.

'

' Coupon rate and yield in per-period decimal format, term in number of complete periods

'

' Price is returned in the same sign as face value

Function bond_price(face_value As Double, coupon_rate As Double, term As Integer, yield As Double)

    If yield = 0 Then

        MsgBox "Error: zero yield in 'bond_price'", vbOKOnly

    Else

        bond_price = face_value * coupon_rate * ((1 - (1 + yield) ^ (-term)) / yield) _

                    + face_value / ((1 + yield) ^ term)

    End If

End Function

' Calculates the price of a bond using a set of forward rates.

'

' Coupon rate in per-period decimal format, term in number of complete periods.

'

' 'rates' should be an array of forward rates, in per-period decimal format. rates(0) is the

' rate used to discount the first coupon. There should be at least 'term' entries in 'rates'.

'

' Price is returned in the same sign as face value.

Function bond_price_fwd(face_value As Double, coupon_rate As Double, term As Integer, rates As Variant)

    Dim i As Integer

    Dim price As Double

    Dim discount_rate As Double

    price = 0

    discount_rate = 1

    For i = 0 To term - 1           ' includes coupon on face value cashflow date

        discount_rate = discount_rate * (1 + rates(i + 1))

        price = price + (face_value * coupon_rate) / discount_rate

    Next

    bond_price_fwd = price + face_value / discount_rate

End Function

' Calculate the duration of a bond, using a set of forward rates.

'

' Coupon rate in per-period decimal format, term in number of complete periods.

'

' 'rates' should be an array of forward rates, in per-period decimal format. rates(0) is the

' rate used to discount the first coupon. There should be at least 'term' entries in 'rates'.

'

' Duration is returned in number of periods (possibly fractional)

Function bond_duration_fwd(face_value As Double, coupon_rate As Double, term As Integer, rates As Variant)

    Dim i As Integer

    Dim price As Double

    Dim discount_rate As Double

    Dim pv As Double

    Dim total As Double

    Dim ok As Integer

    ok = True

    price = bond_price_fwd(face_value, coupon_rate, term, rates)

    discount_rate = 1

    For i = 0 To term - 1               ' includes coupon on face value cashflow date

        If rates(i + 1) = -1 Then

            MsgBox "Error: rate of -100% in 'bond_duration_fwd'", vbOKOnly

            ok = False

        End If

        If ok Then

            discount_rate = discount_rate * (1 + rates(i + 1))

            pv = (face_value * coupon_rate) / discount_rate

            total = total + (pv / price) * CDbl(i + 1)

        End If

    Next

        ' add the pv of the face value to the total and calculate net duration

    If ok And price <> 0 Then

        pv = face_value / discount_rate

        total = total + (pv / price) * CDbl(term)

        bond_duration_fwd = total

    Else

        bond_duration_fwd = 0

    End If

End Function

' Calculates the duration of a bond, using the yield-to-maturity.

'

' Coupon rate in per-period decimal format, term in number of complete periods.

'

' Duration is returned in number of periods (possibly fractional)

Function bond_duration(face_value As Double, coupon_rate As Double, term As Integer, ytm As Double) As Double

    Dim i As Integer

    Dim price As Double

    Dim pv As Double

    Dim total As Double

    If ytm = 0 Then

        MsgBox "Error: zero yield in 'bond_duration'", vbOKOnly

    Else

        price = bond_price(face_value, coupon_rate, term, ytm)

        For i = 0 To term - 1

            pv = (face_value * coupon_rate) / ((1 + ytm) ^ (i + 1))

            total = total + (pv / price) * CDbl(i + 1)

        Next

        pv = face_value / ((1 + ytm) ^ term)

        total = total + (pv / price) * CDbl(term)

    End If

    bond_duration = total

End Function

' Calculates the yield-to-maturity for a speciifed bond price

'

' Coupon rate and yield in per-period decimal format, term in number of complete periods

'

' Yield is returned in per-period decimal format

Function bond_ytm(face_value As Double, coupon_rate As Double, term As Integer, price As Double)

    Dim p1 As Double

    Dim p2 As Double

    Dim y1 As Double

    Dim y2 As Double

    Dim y3 As Double

    Dim attempts As Integer

    Const MAX_ATTEMPTS = 1000

    y1 = 0.1 / term

    y2 = 0.05 / term

    p1 = bond_price(face_value, coupon_rate, term, y1)

    p2 = bond_price(face_value, coupon_rate, term, y2)

    While Abs(price - p2) > 0.00000001 And attempts < MAX_ATTEMPTS

        If p2 = p1 Then

            MsgBox "Error: duplicated points in 'bond_ytm'", vbOKOnly

            attemps = MAX_ATTEMPTS

        Else

            y3 = (price - p1) * ((y2 - y1) / (p2 - p1)) + y1

            y1 = y2

            p1 = p2

            y2 = y3

            p2 = bond_price(face_value, coupon_rate, term, y2)

        End If

        attempts = attempts + 1

        If attempts >= MAX_ATTEMPTS Then

            MsgBox "Failed to locate a solution in 'bond_ytm'", vbOKOnly

        End If

    Wend

    bond_ytm = y2

End Function

